

# NC-Programme im Klartext-Dialog erstellen

# HEIDENHAIN

iTNC 530

TNC 640


TNC 620

TNC 320

```
TNC:\nc_prog\Webinar\PGM.H
0 BEGIN PGM PGM MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+150 Y+100 Z+2
3 ;
4 * - Bezugspunkt aktivieren
5 CYCL DEF 247 BEZUGSPUNKT SETZEN
  Q339=+1 ;BEZUGSPUNKT-NUMMER
6 ;
7 * - Planfraesen T D50
8 TOOL CALL "FACE_MILL_D50" Z S( VC =
1000 )
  FZ0.3
9 M3
10 CYCL DEF 233 PLANFRAESEN
  Q215=+0 ;BEARBEITUNGS-UMFANG
  Q389=+2 ;FRAESSTRATEGIE
  Q350=+1 ;FRAESRICHTUNG
  Q218=+150 ;1. SEITEN-LAENGE
  Q219=+100 ;2. SEITEN-LAENGE
  Q227=+2 ;STARTPUNKT 3. ACHSE
  Q386=+0 ;ENDPUNKT 3. ACHSE
  Q369=+0.3 ;AUFMASS TIEFE
  Q202=+5 ;MAX. ZUSTELL-TIEFE
  Q370=+1 ;BAHN-UEBERLAPPUNG
  Q207= AUTO ;VORSCHUB FRAESEN
```

## Programmablauf

- Erstellung Ordner für das Programm
- Programmanlegen im Klartext-Format
- Definition BLK-Form und Bezugspunkt aktivieren
- Planfräsen T D50
- Kontur Schruppen T D16 mit Aufmaß
- Kontur Schlichten T D16 ohne Aufmaß
- Bohrungen erstellen
  - NC-Anbohrer T D12
  - Bohrer T D6.8
- Kreistasche D20 T D8
- Nut T D8
- Rechtecktasche T D8
- Innenkontur T D8
- Kreistasche D20 Mitte T D8
- Entgraten
- Programm-Ende



Werkstoff: AlCuPbMn





## Erstellung Ordner für das Programm

Neues Verzeichnis

Verzeichnis-Name?

OK ABRUCH

SF: \

TNC: \

- lost+found
- nc\_prog
  - demo
  - Neue\_Zyklen
  - TASTKURS
  - Webinar**
- system
- table
- tncguide

TNC:\nc\_prog\Webinar\\*

Datei-Name	Byte	Status	Datum	Zeit
..			17-02-2015	11:48:12
PGM.H	7483	S +	17-02-2015	11:42:12
PGM_Entgraten.H	4610	E +	17-02-2015	11:48:12

2 Datei(en) 19.32 GByte frei



## Programmanlegen im Klartext-Format

- Cursor auf der rechten Seite (Dateiseite)
- Programmname eingeben
  
- Endung .H für Heidenhain-Klartextprogramm
- Endung .I für DIN-ISO-Programm
- Auswahl mm/Inch für Programmierung

Dialogfenster: Datei wählen

Quell-Datei

PGM.H

OK      ABBRUCH

Dialogfenster: Neue Datei

Maß-System?

TNC:\nc\_prog\Webinar\PGM.H

MM      INCH      ABBRUCH



## Definition BLK-Form

### Mögliche Rohteile

- Rechteckiges Rohteil
- Zylindrisches Rohteil / Rohr
- Rotationssymmetrisches Rohteil mit beliebiger Form




### Aktivierung Bezugspunkt

- Zyklus 247 aktiviert die gewünschte Zeile aus der Preset-Tabelle

### Programmstruktur

- Gliederungspunkt „\* - **Bezugspunkt aktivieren**“  
Tastenkombination SCHIFT + 8
- Kommentar Taste ;
  - Als eigene Zeile
  - Als Kommentar hinter einem Satz

```
0 BEGIN PGM PGM1 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+150 Y+100 Z+2
3 ;
4 * - Bezugspunkt aktivieren
5 CYCL DEF 247 BEZUGSPUNKT SETZEN ~
  Q339=+1 ;BEZUGSPUNKT-NUMMER
6 ;
```





## Planfräsen T D50

Werkzeug aufrufen

- TOOL CALL
- Werkzeug kann mit Nummer oder Name gerufen werden
- Spindelachse **Z**
- Drehzahl ermitteln aus Schnittgeschwindigkeit **vc=1000 m/min**
- **FZ0.3** (Eingabe möglich ab TNC 640 34049x-05 / TNC 620 81760x-02)
- Spindel einschalten mit **M3**

## 7 \* - Planfraesen T D50

```
8 TOOL CALL "FACE_MILL_D50" Z S( VC = 1000 ) ~  
FZ0.3
```

```
9 M3
```

```
10 CYCL DEF 233 PLANFRAESEN ~
```

```
Q215=+0 ;BEARBEITUNGS-UMFANG ~
```

```
Q389=+2 ;FRAESSTRATEGIE ~
```

```
Q350=+1 ;FRAESRICHTUNG ~
```

```
Q218=+150 ;1. SEITEN-LAENGE ~
```

```
Q219=+100 ;2. SEITEN-LAENGE ~
```

```
Q227=+2 ;STARTPUNKT 3. ACHSE ~
```

```
Q386=+0 ;ENDPUNKT 3. ACHSE ~
```

```
Q369=+0.3 ;AUFMASS TIEFE ~
```

```
Q202=+5 ;MAX. ZUSTELL-TIEFE ~
```

```
Q370=+1 ;BAHN-UEBERLAPPUNG ~
```

```
Q207= AUTO ;VORSCHUB FRAESEN ~
```

```
Q385= FZ+0.25 ;VORSCHUB SCHLICHTEN ~
```

```
Q253= MAX ;VORSCHUB VORPOS. ~
```

```
Q357=+2 ;SI.-ABSTAND SEITE ~
```

```
Q200=+2 ;SICHERHEITS-ABST. ~
```

```
Q204=+50 ;2. SICHERHEITS-ABST. ~
```

```
Q347=+0 ;1.BEGRENZUNG ~
```

```
Q348=+0 ;2.BEGRENZUNG ~
```

```
Q349=+0 ;3.BEGRENZUNG ~
```

```
Q220=+0 ;ECKENRADIUS ~
```

```
Q368=+0 ;AUFMASS SEITE ~
```

```
Q338=+0 ;ZUST. SCHLICHTEN
```

```
11 L X+0 Y+0 R0 FMAX M99
```

```
12 ;
```



## Planfräsen T D50

### Planfräsen

- Bearbeitungsumfang **0** = Schruppen und Schlichten mit einem Werkzeug
- Frässtrategie **2** = Fräsbahnen immer in die gleiche Richtung
- 1. und 2. Seitenlänge definiert zu planende Fläche
- Position **X+0 Y+0** anfahren und spanenden Zyklus aufrufen mit **M99**  
Alternativ:
  - L X+0 Y+0 R0 FMAX  
CYCL CALL

→ **M99 / CYCL CALL** ruft den zuletzt definierten spanenden Zyklus auf.

### 7 \* - Planfraesen T D50

```
8 TOOL CALL "FACE_MILL_D50" Z S( VC = 1000 ) ~  
  FZ0.3  
9 M3  
10 CYCL DEF 233 PLANFRAESEN ~  
  Q215=+0 ;BEARBEITUNGS-UMFANG ~  
  Q389=+2 ;FRAESSTRATEGIE ~  
  Q350=+1 ;FRAESRICHTUNG ~  
  Q218=+150 ;1. SEITEN-LAENGE ~  
  Q219=+100 ;2. SEITEN-LAENGE ~  
  Q227=+2 ;STARTPUNKT 3. ACHSE ~  
  Q386=+0 ;ENDPUNKT 3. ACHSE ~  
  Q369=+0.3 ;AUFMASS TIEFE ~  
  Q202=+5 ;MAX. ZUSTELL-TIEFE ~  
  Q370=+1 ;BAHN-UEBERLAPPUNG ~  
  Q207= AUTO ;VORSCHUB FRAESEN ~  
  Q385= FZ+0.25 ;VORSCHUB SCHLICHTEN ~  
  Q253= MAX ;VORSCHUB VORPOS. ~  
  Q357=+2 ;SI.-ABSTAND SEITE ~  
  Q200=+2 ;SICHERHEITS-ABST. ~  
  Q204=+50 ;2. SICHERHEITS-ABST. ~  
  Q347=+0 ;1.BEGRENZUNG ~  
  Q348=+0 ;2.BEGRENZUNG ~  
  Q349=+0 ;3.BEGRENZUNG ~  
  Q220=+0 ;ECKENRADIUS ~  
  Q368=+0 ;AUFMASS SEITE ~  
  Q338=+0 ;ZUST. SCHLICHTEN  
11 L X+0 Y+0 R0 FMAX M99  
12 ;
```





## Kontur Schruppen T D16 mit Aufmaß

### Werkzeugaufwurf mit Aufmaß

- DL, DR und DR2 werden zu den Werten aus der Werkzeugtabelle hinzugerechnet  
→  $L(\text{TOOL.T}) + DL(\text{TOOL.T}) + DL(\text{PGM}) =$   
Tatsächliche Werkzeuglänge
- Kontur wird im Unterprogramm programmiert, da sie mehrfach verwendet wird:
  - Schruppen
  - Schlichten
  - Entgraten
- Unterprogrammaufruf mit CALL LBL 1

```
13 * - Kontur Schruppen T D16
14 TOOL CALL "MILL_D16_ROUGH" Z S9350.353 F3740.141
DL+0.5 DR+0.5
15 M3
16 CALL LBL 1
17 ;
```

```
106 LBL 1
107 L X-25 Y+40 R0 FMAX
108 L Z+5 R0 FMAX
109 L Z-5 R0 F5000
110 APPR LCT X+5 Y+40 R2 RL F AUTO
111 L Y+95
112 RND R8
113 L X+135
114 CR X+145 Y+80 R+20 DR-
115 L Y+24
116 CT Y+9 X+136
117 L X+128.5 Y+5
118 L X+35.5
119 L Y+15
120 CC X+26.5 Y+25
121 C X+17.5 DR+
122 L Y+5
123 L X+5
124 CHF 5
125 L Y+40
126 DEP LCT X-25 Y+40 R2
127 L Z+20 R0 FMAX
128 LBL 0
```



## Kontur Schruppen T D16 mit Aufmaß

### Unterprogramm

- Beginnt mit LBL SET (LBL-Nummer oder –Name)
- Jede LBL-Nummer oder –Name darf nur einmal im NC-Programm vorkommen
- Ende mit LBL 0 (ist immer das Ende des Labels und darf mehrfach im NC-Programm vorkommen)

```
13 * - Kontur Schruppen T D16
14 TOOL CALL "MILL_D16_ROUGH" Z S9350.353 F3740.141
DL+0.5 DR+0.5
15 M3
16 CALL LBL 1
17 ;
```

```
106 LBL 1
107 L X-25 Y+40 R0 FMAX
108 L Z+5 R0 FMAX
109 L Z-5 R0 F5000
110 APPR LCT X+5 Y+40 R2 RL F AUTO
111 L Y+95
112 RND R8
113 L X+135
114 CR X+145 Y+80 R+20 DR-
115 L Y+24
116 CT Y+9 X+136
117 L X+128.5 Y+5
118 L X+35.5
119 L Y+15
120 CC X+26.5 Y+25
121 C X+17.5 DR+
122 L Y+5
123 L X+5
124 CHF 5
125 L Y+40
126 DEP LCT X-25 Y+40 R2
127 L Z+20 R0 FMAX
128 LBL 0
```



## Kontur Schruppen T D16 mit Aufmaß

### Konturbeschreibung:

- Kontur wird mit Klartextbefehlen programmiert:

Taste	Bedeutung	Eingabeparameter
	Line Gerade	■ Koordinaten des Endpunkts
	Chamfer Fase	■ Fasenlänge
	Circle Center Kreismittelpunkt oder Pol	■ Koordinaten des Kreismittelpunkts ■ Koordinaten des Pols
+	+	+
	Circle Kreisbahn um Kreismittelpunkt	■ Koordinaten des Kreisendpunkts ■ Drehrichtung
	Circle Tangential Kreisbahn mit tangentialem Konturanschluss	■ Koordinaten des Kreisendpunkts
	Circle by Radius Kreisbahn mit Radius	■ Koordinaten des Kreisendpunkts ■ Kreisradius ■ Drehrichtung
	Rounding of Corner Ecken runden	■ Eckenradius
	Approach / Departure Kontur anfahren und verlassen	■ Abhängig von der gewählten Funktion
	Freie Konturprogrammierung	■ Bekannte Informationen

```
13 * - Kontur Schruppen T D16
14 TOOL CALL "MILL_D16_ROUGH" Z S9350.353 F3740.141
DL+0.5 DR+0.5
15 M3
16 CALL LBL 1
17 ;
```

```
106 LBL 1
107 L X-25 Y+40 R0 FMAX
108 L Z+5 R0 FMAX
109 L Z-5 R0 F5000
110 APPR LCT X+5 Y+40 R2 RL F AUTO
111 L Y+95
112 RND R8
113 L X+135
114 CR X+145 Y+80 R+20 DR-
115 L Y+24
116 CT Y+9 X+136
117 L X+128.5 Y+5
118 L X+35.5
119 L Y+15
120 CC X+26.5 Y+25
121 C X+17.5 DR+
122 L Y+5
123 L X+5
124 CHF 5
125 L Y+40
126 DEP LCT X-25 Y+40 R2
127 L Z+20 R0 FMAX
128 LBL 0
```



## Kontur Schlichten T D16 ohne Aufmaß

### Werkzeugaufruf ohne Aufmaß

- DL, DR und DR2 können mit Taste NO ENT entfernt werden.
- Unterprogrammaufruf mit CALL LBL 1

```
18 * - Kontur Schlichten T D16
19 TOOL CALL "MILL_D16_ROUGH" Z S9350.353
F3740.141
20 M3
21 CALL LBL 1
22 ;
```

```
106 LBL 1
107 L X-25 Y+40 R0 FMAX
108 L Z+5 R0 FMAX
109 L Z-5 R0 F5000
110 APPR LCT X+5 Y+40 R2 RL F AUTO
111 L Y+95
112 RND R8
113 L X+135
114 CR X+145 Y+80 R+20 DR-
115 L Y+24
116 CT Y+9 X+136
117 L X+128.5 Y+5
118 L X+35.5
119 L Y+15
120 CC X+26.5 Y+25
121 C X+17.5 DR+
122 L Y+5
123 L X+5
124 CHF 5
125 L Y+40
126 DEP LCT X-25 Y+40 R2
127 L Z+20 R0 FMAX
128 LBL 0
```



## Bohrungen erstellen

NC-Anbohrer T D12

- Zentrierzyklus 240
- Angabe D der Senkung -10 mm
- Vorschub über AUTO (Vorschub aus dem TOOL-CALL-Satz wird übernommen)
- Unterprogrammaufruf mit CALL LBL „DRILL“

LBL „DRILL“

- Positionen anfahren und Zyklus mit M99 aufrufen  
Alternativ:
- Bei der **ersten Bohrung M89**
- An jeder weiteren Position wird der zuletzt definierte spanende Zyklus abgearbeitet
- Mit **M99 Automatik beenden**

```
23 * - Bohrungen
24 * - Anbohren T D12
25 TOOL CALL "NC_DEBURRING_D12" Z S2917.841
F291.784
26 M3
27 CYCL DEF 240 ZENTRIEREN ~
  Q200=+2 ;SICHERHEITS-ABST. ~
  Q343=+1 ;AUSWAHL DURCHM/TIEFE ~
  Q201=-10 ;TIEFE ~
  Q344=-10 ;DURCHMESSER ~
  Q206= AUTO ;VORSCHUB TIEFENZ. ~
  Q211=+0 ;VERWEILZEIT UNTEN ~
  Q203=+0 ;KOOR. OBERFLAECHE ~
  Q204=+50 ;2. SICHERHEITS-ABST.
28 CALL LBL "DRILL"
29 ;
```

```
96 LBL "DRILL"
97 L X+12 Y+40 R0 FMAX M99
98 L X+12 Y+55 R0 FMAX M99
99 Q201 = - 22
100 CC X+75 Y+50
101 LP PR+30 PA+45 R0 FMAX M99
102 LP PR+30 IPA+90 R0 FMAX M99
103 LP PR+30 IPA+90 R0 FMAX M99
104 LP PR+30 IPA+90 R0 FMAX M99
105 LBL 0
```



## Bohrungen erstellen

### LBL „DRILL“

#### ■ Verwendung von Polarkoordinaten für Teilkreis

Taste	Bedeutung	Eingabeparameter
	Circle Center Pol	■ Koordinaten des Pols
	Line Polar Geradenbewegung polar	■ Polarkoordinatenradius ■ Polarkoordinatenwinkel
	Circle Polar Kreisbahn polar	■ Polarkoordinatenwinkel ■ Drehsinn
	Circle Tangential Polar Kreisbahn mit tangentialem Konturanchluss polar	■ Polarkoordinatenradius ■ Polarkoordinatenwinkel
	Approach / Departure Kontur anfahren und verlassen	■ Abhängig von der gewählten Funktion

```
23 * - Bohrungen
24 * - Anbohren T D12
25 TOOL CALL "NC_DEBURRING_D12" Z S2917.841
F291.784
26 M3
27 CYCL DEF 240 ZENTRIEREN ~
  Q200=+2 ;SICHERHEITS-ABST. ~
  Q343=+1 ;AUSWAHL DURCHM/TIEFE ~
  Q201=-10 ;TIEFE ~
  Q344=-10 ;DURCHMESSER ~
  Q206= AUTO ;VORSCHUB TIEFENZ. ~
  Q211=+0 ;VERWEILZEIT UNTEN ~
  Q203=+0 ;KOOR. OBERFLAECHE ~
  Q204=+50 ;2. SICHERHEITS-ABST.
28 CALL LBL "DRILL"
29 ;
```

```
96 LBL "DRILL"
97 L X+12 Y+40 R0 FMAX M99
98 L X+12 Y+55 R0 FMAX M99
99 Q201 = - 22
100 CC X+75 Y+50
101 LP PR+30 PA+45 R0 FMAX M99
102 LP PR+30 IPA+90 R0 FMAX M99
103 LP PR+30 IPA+90 R0 FMAX M99
104 LP PR+30 IPA+90 R0 FMAX M99
105 LBL 0
```



## Bohrungen erstellen

Bohrer T D6.8

- Tiefe **-10 mm**
- Über **BEZUG TIEFE = 1** wird die Tiefe auf den zylindrischen Teil des Bohrers bezogen
- Unterprogrammaufruf mit CALL LBL „DRILL“
- Im Unterprogramm wird nach den ersten beiden Bohrungen die Tiefe auf -22 geändert. Dazu den passenden Q-Parameter des Zyklus definieren und anschließend mit M99 aufrufen. Zyklus wird mit geändertem Parameter abgearbeitet.

```
30 * - Bohren T D6.8
31 TOOL CALL "DRILL_D6.8" Z S5149.131 F514.913
32 M3
33 CYCL DEF 200 BOHREN ~
    Q200=+2 ;SICHERHEITS-ABST. ~
    Q201=-10 ;TIEFE ~
    Q206= AUTO ;VORSCHUB TIEFENZ. ~
    Q202=+5 ;ZUSTELL-TIEFE ~
    Q210=+0 ;VERWEILZEIT OBEN ~
    Q203=+0 ;KOOR. OBERFLAECHE ~
    Q204=+50 ;2. SICHERHEITS-ABST. ~
    Q211=+0 ;VERWEILZEIT UNTEN ~
    Q395=+1 ;BEZUG TIEFE
34 CALL LBL "DRILL"
35 ;
```

```
96 LBL "DRILL"
97 L X+12 Y+40 R0 FMAX M99
98 L X+12 Y+55 R0 FMAX M99
99 Q201 = - 22
100 CC X+75 Y+50
101 LP PR+30 PA+45 R0 FMAX M99
102 LP PR+30 IPA+90 R0 FMAX M99
103 LP PR+30 IPA+90 R0 FMAX M99
104 LP PR+30 IPA+90 R0 FMAX M99
105 LBL 0
```



## Kreistasche D20 T D8

### Kreistasche

- Bearbeitungsumfang **0** = Schruppen und Schlichten mit einem Werkzeug
- Kreisdurchmesser
- Aufmaß Seite und Aufmaß Tiefe
- Position (Mitte Kreistasche) **X+125 Y+30** anfahren und spanenden Zyklus aufrufen mit **M99**  
Alternativ:
  - L X+125 Y+30 R0 FMAX  
CYCL CALL

```
36 * - Kreistasche D20 T D8
37 TOOL CALL "MILL_D8_ROUGH" Z S18700.706 F4488.169
38 M3
39 CYCL DEF 252 KREISTASCHE ~
    Q215=+0 ;BEARBEITUNGS-UMFANG ~
    Q223=+20 ;KREISDURCHMESSER ~
    Q368=+0.3 ;AUFMASS SEITE ~
    Q207= AUTO ;VORSCHUB FRAESEN ~
    Q351=+1 ;FRAESART ~
    Q201=-3 ;TIEFE ~
    Q202=+5 ;ZUSTELL-TIEFE ~
    Q369=+0.3 ;AUFMASS TIEFE ~
    Q206=+2805.106 ;VORSCHUB TIEFENZ. ~
    Q338=+0 ;ZUST. SCHLICHTEN ~
    Q200=+2 ;SICHERHEITS-ABST. ~
    Q203=+0 ;KOOR. OBERFLAECHE ~
    Q204=+50 ;2. SICHERHEITS-ABST. ~
    Q370=+1 ;BAHN-UEBERLAPPUNG ~
    Q366=+1 ;EINTAUCHEN ~
    Q385=+2805.106 ;VORSCHUB SCHLICHTEN ~
    Q439=+0 ;BEZUG VORSCHUB
40 L X+125 Y+30 R0 FMAX M99
41 ;
```





## Nut T D8

### Gerade Nut

- Abmaße Nut  
Länge 32 mm (20 mm + Nutbreite 12 mm) Breite 12 mm
- Drehlage eingeben 120°
- Nutlage = 2 (bemaßte Position der Nut → diese Position muss vor Aufruf angefahren werden)
- Position (Nutlage) **X+130 Y+65** anfahren und spanenden Zyklus aufrufen mit **M99**  
Alternativ:
  - L X+130 Y+65 R0 FMAX  
CYCL CALL

```
42 * - Nut T D8
43 CYCL DEF 253 NUTENFRAESEN ~
  Q215=+0 ;BEARBEITUNGS-UMFANG ~
  Q218=+32 ;NUTLAENGE ~
  Q219=+12 ;NUTBREITE ~
  Q368=+0.2 ;AUFMASS SEITE ~
  Q374=+120 ;DREHLAGE ~
  Q367=+2 ;NUTLAGE ~
  Q207= AUTO ;VORSCHUB FRAESEN ~
  Q351=+1 ;FRAESART ~
  Q201=-6 ;TIEFE ~
  Q202=+3 ;ZUSTELL-TIEFE ~
  Q369=+0.2 ;AUFMASS TIEFE ~
  Q206=+2805.106 ;VORSCHUB TIEFENZ. ~
  Q338=+0 ;ZUST. SCHLICHTEN ~
  Q200=+2 ;SICHERHEITS-ABST. ~
  Q203=+0 ;KOOR. OBERFLAECHE ~
  Q204=+50 ;2. SICHERHEITS-ABST. ~
  Q366=+2 ;EINTAUCHEN ~
  Q385=+2805.106 ;VORSCHUB SCHLICHTEN ~
  Q439=+3 ;BEZUG VORSCHUB
44 L X+130 Y+65 R0 FMAX M99
45 ;
```



## Rechtecktasche T D8

### Rechtecktasche

- Abmaße Tasche
  1. Seitenlänge 30 mm (X-Achse)
  2. Seitenlänge 20 mm (Y-Achse)
- Eckenradius eingeben 5 mm (bei 0 wird WKZ-Radius verwendet)
- Taschenlage = 0 (bemaßte Position der Tasche → diese Position muss vor Aufruf angefahren werden)
- Position (Taschenlage) **X+29 Y+71** anfahren und spanenden Zyklus aufrufen mit **M99**  
Alternativ:
- L X+29 Y+71 R0 FMAX  
CYCL CALL

```
46 * - Rechtecktasche T D8
47 CYCL DEF 251 RECHTECKTASCHE ~
  Q215=+0 ;BEARBEITUNGS-UMFANG ~
  Q218=+30 ;1. SEITEN-LAENGE ~
  Q219=+20 ;2. SEITEN-LAENGE ~
  Q220=+5 ;ECKENRADIUS ~
  Q368=+0.3 ;AUFMASS SEITE ~
  Q224=+45 ;DREHLAGE ~
  Q367=+0 ;TASCHENLAGE ~
  Q207= AUTO ;VORSCHUB FRAESEN ~
  Q351=+1 ;FRAESART ~
  Q201=-5 ;TIEFE ~
  Q202=+5 ;ZUSTELL-TIEFE ~
  Q369=+0.2 ;AUFMASS TIEFE ~
  Q206=+2805.106 ;VORSCHUB TIEFENZ. ~
  Q338=+0 ;ZUST. SCHLICHTEN ~
  Q200=+2 ;SICHERHEITS-ABST. ~
  Q203=+0 ;KOOR. OBERFLAECHE ~
  Q204=+50 ;2. SICHERHEITS-ABST. ~
  Q370=+1 ;BAHN-UEBERLAPPUNG ~
  Q366=+1 ;EINTAUCHEN ~
  Q385=+2805.106 ;VORSCHUB SCHLICHTEN
48 L X+29 Y+71 R0 FMAX M99
49 ;
```



## Innenkontur T D8

### SL-Zyklen

- Konturbeschreibung in Unterprogrammen
- Zyklus 14: Zuweisung der Unterprogramme
- Zyklus 20: Konturdaten
  - Tiefe
  - Aufmaße
- Zyklus 22: Ausräumen
  - Technologiedaten
- Werkzeug vorpositionieren **X+0 Y+0** und spanenden Zyklus aufrufen mit **M99**

→ Zyklus 14 und Zyklus 20 erzeugen keine Späne und müssen nicht aufgerufen werden (DEF-Aktive Zyklen)

```
50 * - Innenkontur T D8
51 CYCL DEF 14.0 KONTUR
52 CYCL DEF 14.1 KONTURLABEL10
53 CYCL DEF 20 KONTUR-DATEN ~
    Q1=-5 ;FRAESTIEFE ~
    Q2=+1 ;BAHN-UEBERLAPPUNG ~
    Q3=+0.2 ;AUFMASS SEITE ~
    Q4=+0.2 ;AUFMASS TIEFE ~
    Q5=+0 ;KOOR. OBERFLAECHE ~
    Q6=+2 ;SICHERHEITS-ABST. ~
    Q7=+50 ;SICHERE HOEHE ~
    Q8=+0.5 ;RUNDUNGSRADIUS ~
    Q9=+1 ;DREHSINN
54 CYCL DEF 22 AUSRAEUMEN ~
    Q10=-5 ;ZUSTELL-TIEFE ~
    Q11=+2805.106 ;VORSCHUB TIEFENZ. ~
    Q12= AUTO ;VORSCHUB RAEUMEN ~
    Q18=+0 ;VORRAEUM-WERKZEUG ~
    Q19=+2805.106 ;VORSCHUB PENDELN ~
    Q208= MAX ;VORSCHUB RUECKZUG ~
    Q401=+60 ;VORSCHUBFAKTOR ~
    Q404=+0 ;NACHRAEUMSTRATEGIE
55 L X+0 Y+0 Z+50 R0 FMAX M99
```



## Innenkontur T D8

### Konturbeschreibung

- Nullpunktverschiebung auf Mitte Teil erspart Rechenarbeit
- Reine Kontur in X/Y
- Keine Anfahrbewegung/M-Funktionen/Vorschübe/Z-Verfahrbewegung
- Konturbeschreibung z.B. mit grauen Bahnfunktionstasten (kartesische und polare Koordinaten) sowie FK-Programmierung (Freie Konturprogrammierung)

```
67 LBL 10
68 CYCL DEF 7.0 NULLPUNKT
69 CYCL DEF 7.1 X+75
70 CYCL DEF 7.2 Y+50
71 L X-25 Y-10 RR
72 CR Y+10 R+10 DR-
73 L X-10
74 RND R3
75 L Y+25
76 CT X+10
77 L Y+10
78 RND R3
79 L X+25
80 CT Y-10
81 L X+10
82 RND R3
83 L Y-25
84 CT X-10
85 L Y-10
86 RND R3
87 L X-25
88 CYCL DEF 7.0 NULLPUNKT
89 CYCL DEF 7.1 X+0
90 CYCL DEF 7.2 Y+0
91 CYCL DEF 7.3 Z+0
92 LBL 0
```



## Innenkontur T D8

### SL-Zyklen Schlichten

- Definierte Aufmaße aus dem Zyklus 20 werden geschlichtet
- Werkzeug vorpositionieren **X+0 Y+0** und spanenden Zyklus aufrufen mit **M99**

```
56 CYCL DEF 23 SCHLICHTEN TIEFE ~
   Q11=+2805.106 ;VORSCHUB TIEFENZ. ~
   Q12=+2805.106 ;VORSCHUB RAEUMEN ~
   Q208= MAX ;VORSCHUB RUECKZUG
57 L X+0 Y+0 Z+50 R0 FMAX M99
58 CYCL DEF 24 SCHLICHTEN SEITE ~
   Q9=+1 ;DREHSINN ~
   Q10=-5 ;ZUSTELL-TIEFE ~
   Q11=+2805.106 ;VORSCHUB TIEFENZ. ~
   Q12=+2805.106 ;VORSCHUB RAEUMEN ~
   Q14=+0 ;AUFMASS SEITE
59 L X+0 Y+0 Z+50 R0 FMAX M99
60 ;
```



## Kreistasche D20 T D8

### Kreistasche

- Vertiefter Start bei Z-5 mm  
→ Koordinatenoberfläche
- Folgende Eingaben beziehen sich auf die Koordinatenoberfläche:
  - Tiefe
  - Sicherheits-Abstand
  - 2. Sicherheitsabstand
- Position (Mitte Kreistasche) **X+75 Y+50** anfahren und spanenden Zyklus aufrufen mit **M99**

```
61 * - Kreistasche D20 Mitte T D8
62 CYCL DEF 252 KREISTASCHE ~
  Q215=+0 ;BEARBEITUNGS-UMFANG ~
  Q223=+20 ;KREISDURCHMESSER ~
  Q368=+0.3 ;AUFMASS SEITE ~
  Q207= AUTO ;VORSCHUB FRAESEN ~
  Q351=+1 ;FRAESART ~
  Q201=-3 ;TIEFE ~
  Q202=+5 ;ZUSTELL-TIEFE ~
  Q369=+0.3 ;AUFMASS TIEFE ~
  Q206=+2805.106 ;VORSCHUB TIEFENZ. ~
  Q338=+0 ;ZUST. SCHLICHTEN ~
  Q200=+2 ;SICHERHEITS-ABST. ~
  Q203=-5 ;KOOR. OBERFLAECHE ~
  Q204=+50 ;2. SICHERHEITS-ABST. ~
  Q370=+1 ;BAHN-UEBERLAPPUNG ~
  Q366=+1 ;EINTAUCHEN ~
  Q385=+2805.106 ;VORSCHUB SCHLICHTEN ~
  Q439=+0 ;BEZUG VORSCHUB
63 L X+75 Y+50 R0 FMAX M99
64 ;
```



## Entgraten

### Programm Entgraten

- Bestehendes Programm kopieren
- Werkzeug NC-Anbohrer D8
  - L auf Spitze vermessen
  - R = Außendurchmesser / 2
- Wird mit DL-1 und DR-3 korrigiert
- In den Zyklen nur Schlichten und nur Aufmaß Seite definieren
- Tiefe gibt die Größe der Fase an 0.5 mm

→ Die Simulation zeigt die Fase an, wenn man in der Spalte T-Angle den Spitzenwinkel des Werkzeuges hinterlegt hat (z. B. 90°)


65 CALL PGM TNC:\nc\_prog\Webinar\PGM\_Entgraten.H

```
1 * Entgraten
2 * - Kontur Schlichten T D16
3 TOOL CALL "NC_SPOT_DRILL_D8" Z S9350.353
F3740.141 DL-1 DR-3
4 M3
5 CALL LBL 1
6 ;
7 * - Kreistasche D20 T D8
8 M3
9 CYCL DEF 252 KREISTASCHE ~
  Q215=+2 ;BEARBEITUNGS-UMFANG ~
  Q223=+20 ;KREISDURCHMESSER ~
  Q368=+0.3 ;AUFMASS SEITE ~
  Q207= AUTO ;VORSCHUB FRAESEN ~
  Q351=+1 ;FRAESART ~
  Q201=-0.5 ;TIEFE ~
  Q202=+5 ;ZUSTELL-TIEFE ~
  Q369=+0 ;AUFMASS TIEFE ~
  Q206=+2805.106 ;VORSCHUB TIEFENZ. ~
  Q338=+0 ;ZUST. SCHLICHTEN ~
  Q200=+2 ;SICHERHEITS-ABST. ~
  ....
  Q385=+2805.106 ;VORSCHUB SCHLICHTEN ~
  Q439=+0 ;BEZUG VORSCHUB
10 L X+125 Y+30 R0 FMAX M99
11 ;
```



# Simulation

MW M-TS/ Feb 2015








# Simulation mit Werkzeugbahnen

MW M-TS/ Feb 2015





## Programm-Ende

66 M30

Programm beenden:

- M2/M30 beendet das Hauptprogramm
- Hinter M2/M30 können Unterprogramme stehen  
→ Diese werden dann nicht nochmal gelesen
- Vor Programmende kann z. B.
  - Eine sichere Position angefahren werden
  - Das Werkzeug aus der Spindel mit  
TOOL CALL 0 gewechselt werden

# **Wechsel zum Programmierplatz**